Chapter 23
Rainbow

23.1 Statement of the Problem

We consider scattering of light by a spherical water drop in geometrical optics. For small drops, diffraction becomes significant; our analysis is only valid for drops which are not too small. Let the drop radius be 1. The ray with the impact parameter ρ splits into the reflected ray and the refracted one. Their directions are given by the Snell law

```
In[1] := Snell = {α−>ArcSin[ρ], β−>ArcSin[ρ/n]};
```

where the refraction index of water is

```
In[2] := Water = n−>1.333;
In[3] := col = {RGBColor[1,0,0], RGBColor[0,0,1], RGBColor[0,1,0],
      RGBColor[1,0,1], RGBColor[0,1,1], RGBColor[1,1,0]};
In[4] := With[{y = 0.6},
   With[{α = α/.Snell/.ρ−>y, β = β/.Snell/.ρ−>y/.Water,
       x = −Sqrt[1 − y^2]},
     With[{φ = π − α, ϑ = π − 2∗α}, With[{ψ = φ + 2∗β − π},
       Graphics[{Black, Circle[], Line[{{−2,0}, {1.2,0}}],
         Line[{{0,0}, {2∗Cos[φ], 2∗Sin[φ]}}],
         Line[{{0,0}, {Cos[ψ], Sin[ψ]}}],
         Line[{{x,y}, {x,0}}], col[[1]], Line[{{−2,y}, {x,y}}],
         col[[2]], Line[{{x,y}, {x + 1.5∗Cos[ϑ], y + 1.5∗Sin[ϑ]}}],
         col[[3]], Line[{{x,y}, {Cos[ψ], Sin[ψ]}}],
         Black, Inset[Style["α", 24], {−0.25, 0.1}],
         Inset[Style["α", 24], {x − 0.25, y + 0.1}],
         Inset[Style["α", 24], {x − 0.15, y + 0.25}],
         Inset[Style["β", 24], {x + 0.35, y − 0.15}],
         Inset[Style["ρ", 24], {x + 0.1, 0.5∗y}]}]]]]
```

A. Grozin, *Introduction to Mathematica® for Physicists*, Graduate Texts in Physics,
DOI 10.1007/978-3-319-00894-3_23,

Out[4] =



The incident ray hits the drop at the point $\{\mathrm{Cos}[\varphi], \mathrm{Sin}[\varphi]\}$, where $\varphi = \pi - \alpha$. The reflected ray has the direction $\vartheta = \pi - 2\alpha$. The refracted ray hits the drop surface again at $\{\mathrm{Cos}[\psi], \mathrm{Sin}[\psi]\}$, where $\psi = \varphi + 2\beta - \pi$.

The incident light has two polarizations, with the electric field orthogonal to the scattering plane or lying in this plane. The reflection coefficients for these polarizations are given by the Fresnel formulas [21] (they don't depend on the direction, i.e., are the same for a ray entering water and a ray leaving it).

```
In[5] := Rs = (Sin[α − β]/Sin[α + β])^2; Rp = (Tan[α − β]/Tan[α + β])^2;
```

23.2 0 Ray Segments Inside the Drop

First let's consider rays reflected by the drop immediately after they hit its surface.

```
In[6] := Ray0[y_, a_] := With[{α = α/. Snell/. ρ−>y,
            β = β/. Snell/. ρ−>y/. Water, x = −Sqrt[1 − y^2]},
      With[{ϑ = π − 2 ∗ α}, {col[[1]], Line[{{−1 − a, y}, {x, y}}],
            col[[2]], Line[{{x, y}, {x + a ∗ Cos[ϑ], y + a ∗ Sin[ϑ]}}]}]]
In[7] := Manipulate[Graphics[Join[{Black, Circle[]}, Ray0[y, a]],
            PlotRange−>{{−1 − a, 1.1}, {−1.1, a + 0.7}}],
      {{y, 0.6}, 0, 1}, {{a, 2}, 1, 10}]
```

Out[7] =




```
In[8] := Manipulate[
  Graphics[
    Join[{Black, Circle[]},
      With[{δ = (ρmax − ρmin)/M},
        If[δ > 0, Apply[Join, Table[Ray0[y, a], {y, ρmin + δ/2, ρmax, δ}]], {}]]],
    PlotRange−>{{−1 − a, 1.1}, {−1.1, a + 0.7}}],
  {{ρmin, 0}, 0, 1}, {{ρmax, 1}, 0, 1},
  {{M, 10}, Table[i, {i, 30}]}, {{a, 2}, 1, 10}]
```

Out[8] =



The scattering angle is

In[9] := $\vartheta 0 = \pi - 2 * \alpha$;

In[10] := Plot[Evaluate[ϑ0/. Snell], {ρ, 0, 1}]

Out[10] =



In order to calculate the differential cross section, we need to express ρ via the scattering angle ϑ:

In[11] := sol0 = Simplify[Solve[(ϑ0/.Snell) == ϑ, ρ], 0 < ϑ < π]

$\text{Out}[11] = \left\{\left\{\rho \to \text{Cos}\left[\frac{\vartheta}{2}\right]\right\}\right\}$

In[12] := ρ0 = ρ/.sol0[[1]]

$\text{Out}[12] = \text{Cos}\left[\frac{\vartheta}{2}\right]$

In[13] := Clear[sol0]

The angles α and β are

In[14] := Snell0 = {α−>(π − ϑ)/2, β−>ArcSin[Sin[α]/n]};

The area of the ring in the transverse plane corresponding to the scattering angles between ϑ and $\vartheta + \mathrm{d}\vartheta$, divided by $\mathrm{d}\Omega = 2\pi\,\text{Sin}[\vartheta]\,\mathrm{d}\vartheta$, is

In[15] := σ0 = Simplify[−D[ρ0^2, ϑ]/(2 ∗ Sin[ϑ])]

$\text{Out}[15] = \frac{1}{4}$

Therefore, the cross sections for the two polarizations are

In[16] := σ0s[ϑ_] = Simplify[σ0 ∗ Rs//.Snell0/.Water];
σ0p[ϑ_] = Simplify[σ0 ∗ Rp//.Snell0/.Water];

In[17] := Plot[{σ0s[ϑ], σ0p[ϑ], (σ0s[ϑ] + σ0p[ϑ])/2}, {ϑ, 0, π},
PlotRange−>All, PlotStyle−>col]

Out[17] =



Note that there is a scattering angle ϑ at which the scattered light is completely polarized: its electric field is orthogonal to the scattering plane. This happens when α is equal to the Brewster angle αB

In[18] := αB = ArcTan[n]; αB/.Water

Out[18] = 0.927175

In this case

In[19] := βB = ArcTan[1/n];

so that $\alpha\text{B} + \beta\text{B} = \pi/2$ —the refracted ray is perpendicular to the reflected one:

**In[20] := Graphics[{Line[{{−1,0},{1,0}}],Line[{{0,−1},{0,1}}],
col[[1]],Line[{{0,0},{−1,1/*n*}}],
col[[2]],Line[{{0,0},{1,1/*n*}}],
col[[3]],Line[{{0,0},{1/*n*,−1}}],
Black,Inset[Style["*α*",24],{−0.06,0.13}],Inset[Style["*α*",24],{0.06,0.13}],
Inset[Style["*β*",24],{0.06,−0.13}]}]/. Water**

Out[20] =



When the incoming light with the electric field in the scattering plane reaches water, electric dipoles in it oscillate along the direction perpendicular to the refracted ray; they don't radiate in the direction along this axis, i.e., don't produce the reflected ray: Rp = 0.

In[21] := Plot[σ0p[ϑ],{ϑ,0,π},PlotRange−>{0,0.01}]

Out[21] =



In[22] := ϑ0/. α−>αB/. Water

Out[22] = 1.28724

23.3 1 Ray Segment Inside the Drop

```
In[23] := Ray1[y_,a_] := With[{α = α/.Snell/.ρ->y,
        β = β/.Snell/.ρ->y/.Water,x = −Sqrt[1 − y^2]},
    With[{φ = π − α, ϑ = 2 ∗ (β − α)}, With[{ψ = φ + 2 ∗ β − π},
        With[{x1 = Cos[ψ], y1 = Sin[ψ]}, {col[[1]], Line[{{−1 − a, y}, {x, y}}],
            col[[3]], Line[{{x, y}, {x1, y1}}],
            col[[2]], Line[{{x1, y1}, {x1 + a ∗ Cos[ϑ], y1 + a ∗ Sin[ϑ]}}]}]]]]
In[24] := Manipulate[Graphics[Join[{Black, Circle[]}, Ray1[y, a]],
        PlotRange->{{−1 − a, 1 + a}, {−1.1, 1.1}}], {{y, 0.6}, 0, 1},
    {{a, 2}, 1, 10}]
```

Out[24] =



When entering water, the ray is deflected by $\alpha - \beta$ clockwise; when leaving water, it is deflected by the same angle again. The direction of the outgoing ray is

```
In[25] := ϑ1 = 2 ∗ (β − α);
In[26] := Manipulate[
  Graphics[
    Join[{Black, Circle[]},
      With[{δ = (ρmax − ρmin)/M},
        If[δ > 0, Apply[Join, Table[Ray1[y, a], {y, ρmin + δ/2, ρmax, δ}]], {}]]],
    PlotRange->{{−1 − a, 1 + a}, {−1.1, 1.1}}],
  {{ρmin, 0}, 0, 1}, {{ρmax, 1}, 0, 1},
  {{M, 10}, Table[i, {i, 30}]}, {{a, 2}, 1, 10}]
```

Out[26] =



The scattering angle ϑ is obtained by reducing $\vartheta 1$ to the interval $[-\pi, \pi]$ and then taking Abs; in the present case, it is just $-\vartheta 1$.

In[27] := Plot[Evaluate[$-\vartheta$1/. Snell/. Water], {ρ, 0, 1}]

Out[27] =



It varies from 0 to

In[28] := ϑ1m = Simplify[$-\vartheta$1/. Snell/. ρ -> 1]

Out[28] = $\pi - 2\,\mathrm{ArcSin}\left[\frac{1}{n}\right]$

In[29] := ϑ1m/. Water

Out[29] = 1.4449

Now we have to solve the equation $\alpha - \beta = \vartheta/2$ for ρ.

In[30] := eq1 = (TrigExpand[Sin[$\alpha - \beta$]]/. {Sin[α] -> ρ, Cos[α] -> Sqrt[1 − ρ^2], Sin[β] -> ρ/n, Cos[β] -> Sqrt[1 − (ρ/n)^2]}) == Sin[ϑ/2]

Out[30] = $-\frac{\rho\sqrt{1-\rho^2}}{n} + \rho\sqrt{1-\frac{\rho^2}{n^2}} == \mathrm{Sin}\left[\frac{\vartheta}{2}\right]$

In[31] := sol1 = Solve[eq1, ρ]

Out[31] = $\left\{\left\{\rho \to -\frac{\sqrt{n^2 - n^2\mathrm{Cos}[\vartheta]}}{\sqrt{2}\sqrt{1+n^2-2n\,\mathrm{Cos}\left[\frac{\vartheta}{2}\right]}}\right\}, \left\{\rho \to \frac{\sqrt{n^2 - n^2\mathrm{Cos}[\vartheta]}}{\sqrt{2}\sqrt{1+n^2-2n\,\mathrm{Cos}\left[\frac{\vartheta}{2}\right]}}\right\}, \left\{\rho \to -\frac{\sqrt{n^2 - n^2\mathrm{Cos}[\vartheta]}}{\sqrt{2}\sqrt{1+n^2+2n\,\mathrm{Cos}\left[\frac{\vartheta}{2}\right]}}\right\}, \left\{\rho \to \frac{\sqrt{n^2 - n^2\mathrm{Cos}[\vartheta]}}{\sqrt{2}\sqrt{1+n^2+2n\,\mathrm{Cos}\left[\frac{\vartheta}{2}\right]}}\right\}\right\}$

We discard the negative solutions. The positive ones evaluated at ϑ1m are

In[32] := Simplify[ρ/. sol1[[{2,4}]]/. ϑ -> ϑ1m, $n > 1$]

Out[32] = $\left\{1, \sqrt{\frac{-1+n^2}{3+n^2}}\right\}$

So, the right solution is number 2:

**In[33] := ρ1 = Simplify[Simplify[ρ/. sol1[[2]], {n > 1, 0 < ϑ < π}]/.
Cos[ϑ]−>1 − 2 ∗ Sin[ϑ/2]^2, {n > 1, ϑ > 0, ϑ < π}]**

$$\text{Out[33]} = \frac{n\,\text{Sin}\left[\frac{\vartheta}{2}\right]}{\sqrt{1+n^2-2n\,\text{Cos}\left[\frac{\vartheta}{2}\right]}}$$

In[34] := Clear[eq1, sol1]

The geometrical cross section for ϑ between ϑ and $\vartheta + d\vartheta$, divided by $d\Omega$, is

**In[35] := σ1 = Simplify[D[ρ1^2, ϑ]/(4 ∗ c2 ∗ s2)/.
{Cos[ϑ/2]−>c2, Sin[ϑ/2]−>s2}]**

$$\text{Out[35]} = \frac{n^2\left(\text{c2} - 2\,\text{c2}^2 n + \text{c2}\,n^2 - n\,\text{s2}^2\right)}{4\,\text{c2}\left(1 - 2\,\text{c2}\,n + n^2\right)^2}$$

where $\text{c2} = \text{Cos}[\vartheta/2]$, $\text{s2} = \text{Sin}[\vartheta/2]$. The angles α and β are

In[36] := Snell1 = {α−>ArcSin[ρ1], β−>ArcSin[ρ1/n]};

The differential cross sections for the two polarizations are

In[37] := cs2 = {c2−>Cos[ϑ/2], s2−>Sin[ϑ/2]};

**In[38] := σ1s[ϑ_] = Simplify[σ1 ∗ (1 − Rs)^2/. Snell1/. cs2/. Water];
σ1p[ϑ_] = Simplify[σ1 ∗ (1 − Rp)^2/. Snell1/. cs2/. Water];**

(the transmission coefficient is $T = 1 - R$, and transmission happens twice).

**In[39] := Plot[{σ1s[ϑ], σ1p[ϑ], (σ1s[ϑ] + σ1p[ϑ])/2}, {ϑ, 0, ϑ1m/. Water},
PlotRange−>All, PlotStyle−>col]**

Out[39] =



23.4 2 Ray Segments Inside the Drop

**In[40] := Ray2[y_, a_] := With[{α = α/. Snell/. ρ−>y,
β = β/. Snell/. ρ−>y/. Water, x = −Sqrt[1 − y^2]},
With[{ϑ = 4 ∗ β − 2 ∗ α − π},
Module[{φ = π − α, R = {col[[1]], Line[{{−1 − a, y}, {x, y}}]},
x1 = x, y1 = y, x2, y2, ψ},**

```
        ψ = φ + 2 ∗ β − π; x2 = Cos[ψ]; y2 = Sin[ψ];
        R = Join[R, {col[[3]], Line[{{x1, y1}, {x2, y2}}]}];
        φ = ψ; x1 = x2; y1 = y2; ψ = φ + 2 ∗ β − π; x2 = Cos[ψ]; y2 = Sin[ψ];
        R = Join[R, {col[[4]], Line[{{x1, y1}, {x2, y2}}]}];
        Join[R, {col[[2]], Line[{{x2, y2}, {x2 + a ∗ Cos[ϑ], y2 + a ∗ Sin[ϑ]}}]}]]]]
In[41] := Manipulate[Graphics[Join[{Black, Circle[]}, Ray2[y, a]],
          PlotRange−>{{−1 − a, 1.1}, {−1 − 0.7 ∗ a, 1.1}}],
    {{y, 0.6}, 0, 1}, {{a, 2}, 1, 10}]
```

Out[41] =



The second segment of the ray inside the drop is obtained from the first one by rotating by the angle $\pi - 2\beta$ clockwise; hence the outgoing ray is obtained from the one in the previous section by the same rotation:

```
In[42] := ϑ2 = 4 ∗ β − 2 ∗ α − π;
In[43] := Manipulate[
   Graphics[
     Join[{Black, Circle[]},
       With[{δ = (ρmax − ρmin)/M},
         If[δ > 0, Apply[Join, Table[Ray2[y, a], {y, ρmin + δ/2, ρmax, δ}]], {}]]],
     PlotRange−>{{−1 − a, 1.1}, {−1 − 0.7 ∗ a, 1.1}}],
   {{ρmin, 0}, 0, 1}, {{ρmax, 1}, 0, 1},
   {{M, 10}, Table[i, {i, 30}]}, {{a, 2}, 1, 10}]
```

Out[43] =



The real scattering angle is $\vartheta = -\vartheta 2$:

In[44] := Plot[−ϑ2/. Snell/. Water, {ρ, 0, 1}]

Out[44] =



It has a minimum at

In[45] := s2r = Solve[D[ϑ2/. Snell, ρ] == 0, ρ]

$$\text{Out[45]} = \left\{\left\{\rho \to -\frac{\sqrt{4-n^2}}{\sqrt{3}}\right\}, \left\{\rho \to \frac{\sqrt{4-n^2}}{\sqrt{3}}\right\}\right\}$$

In[46] := ρ2r = ρ/. s2r[[2]]

$$\text{Out[46]} = \frac{\sqrt{4-n^2}}{\sqrt{3}}$$

equal to

In[47] := ϑ2r = −ϑ2/. Snell/. ρ−>ρ2r

$$\text{Out[47]} = \pi + 2\,\text{ArcSin}\left[\frac{\sqrt{4-n^2}}{\sqrt{3}}\right] - 4\,\text{ArcSin}\left[\frac{\sqrt{4-n^2}}{\sqrt{3}n}\right]$$

In[48] := {ρ2r, ϑ2r}/. Water

Out[48] = {0.860835, 2.40719}

Rays from a relatively wide ring around $\rho 2\text{r}$ have practically the same scattering angle $\vartheta 2\text{r}$. This contribution to the cross section tends to ∞ at this angle. When an observer sees a cloud of water drops illuminated by the sun, especially bright light rays arrive along the cone with angle $\pi - \vartheta 2\text{r}$. Usually, only a part of the circle is seen (the full circle can be sometimes observed from an airplane).

```
In[49] := With[{R = −Tan[ϑ2r/. Water]}, Graphics3D[
  Join[{Opacity[0.1], Yellow, Cone[{{1,0,0}, {0,0,0}}, R]},
    Apply[Join, Table[
      Module[{x = 0.5 ∗ (Random[] + 1), φ = 2 ∗ π ∗ Random[], y, z},
        y = R ∗ x ∗ Cos[φ]; z = R ∗ x ∗ Sin[φ];
        {Red, Line[{{0, y, z}, {x, y, z}}], Blue, Line[{{x, y, z}, {0,0,0}}]}],
      {n, 50}]]],
  Boxed−>False, ViewPoint−>{−10, −30, 0}]]
```

Out[49] =



In[50] := Show[Import["rainbow.jpg"]]

Out[50] = 

The scattering angle at $\rho = 1$ is

In[51] := ϑ2m = −ϑ2/. Snell/. ρ−>1

$\text{Out}[51] = 2\pi - 4\,\text{ArcSin}\left[\frac{1}{n}\right]$

In[52] := {ϑ2rw, ϑ2mw} = {ϑ2r, ϑ2m}/. Water;

Now we have to solve the equation $\alpha - 2\beta = (\vartheta - \pi)/2$ for ρ.

In[53] := eq2 = (TrigExpand[Sin[α − 2 ∗ β]]/.
{Sin[α] → ρ, Cos[α] → Sqrt[1 − ρ^2],
Sin[β] → ρ/n, Cos[β] → Sqrt[1 − (ρ/n)^2]}) == −Cos[ϑ/2]

$$\text{Out}[53] = -\frac{\rho^3}{n^2} - \frac{2\rho\sqrt{1-\rho^2}\sqrt{1-\frac{\rho^2}{n^2}}}{n} + \rho\left(1 - \frac{\rho^2}{n^2}\right) == -\text{Cos}\left[\frac{\vartheta}{2}\right]$$

In[54] := sol2 = Solve[eq2, ρ];

The solution number 3 is the smaller one; number 4 is the larger one (1 and 2 are negative).

In[55] := ρ2a = ρ/. sol2[[3]]; ρ2b = ρ/. sol2[[4]];

In[56] := Clear[eq2, sol2]

In[57] := p2a = ParametricPlot[{ϑ, ρ2a/. Water}, {ϑ, ϑ2r/. Water, π},
PlotStyle → Blue];
p2b = ParametricPlot[{ϑ, ρ2b/. Water}, {ϑ, ϑ2r/. Water, ϑ2m/. Water},
PlotStyle → Red];
Show[p2a, p2b, PlotRange → {{ϑ2r/. Water, π}, {0, 1}}]

Out[57] =



In order to scatter between ϑ and $\vartheta + \mathrm{d}\vartheta$, the incident ray has to hit one of the two rings, σ2a or (if $\vartheta < \vartheta$2m) σ2b.

**In[58] := σ2a = −(*D*[ρ2a^2, ϑ]/(4 ∗ c2 ∗ s2)/. {Cos[ϑ/2] → c2, Sin[ϑ/2] → s2,
Cos[ϑ] → c2^2 − s2^2, Sin[ϑ]−>2 ∗ c2 ∗ s2});
σ2b = *D*[ρ2b^2, ϑ]/(4 ∗ c2 ∗ s2)/. {Cos[ϑ/2] → c2, Sin[ϑ/2] → s2,
Cos[ϑ] → c2^2 − s2^2, Sin[ϑ]−>2 ∗ c2 ∗ s2};**

The angles α and β in these two cases are

**In[59] := Snell2a = {α−>ArcSin[ρ2a], β−>ArcSin[ρ2a/*n*]};
Snell2b = {α−>ArcSin[ρ2b], β−>ArcSin[ρ2b/*n*]};**

The differential cross sections for the two polarizations are

**In[60] := σ2as[ϑ_] = σ2a ∗ (1 − Rs)^2 ∗ Rs/. Snell2a/. cs2/. Water;
σ2ap[ϑ_] = σ2a ∗ (1 − Rp)^2 ∗ Rp/. Snell2a/. cs2/. Water;
σ2bs[ϑ_] = σ2b ∗ (1 − Rs)^2 ∗ Rs/. Snell2b/. cs2/. Water;
σ2bp[ϑ_] = σ2b ∗ (1 − Rp)^2 ∗ Rp/. Snell2b/. cs2/. Water;**

**In[61] := σ2s[ϑ_] := If[ϑ > ϑ2rw, σ2as[ϑ] + If[ϑ < ϑ2mw, σ2bs[ϑ], 0], 0];
σ2p[ϑ_] := If[ϑ > ϑ2rw, σ2ap[ϑ] + If[ϑ < ϑ2mw, σ2bp[ϑ], 0], 0];**

```
In[62] := Plot[{σ2s[ϑ], σ2p[ϑ], (σ2s[ϑ] + σ2p[ϑ])/2}, {ϑ, ϑ2rw + 0.001, π},
    PlotRange->{0, 1}, PlotStyle->col]
```

Out[62] =



The considered contributions (0, 1, 2 ray segments inside the drop) to the cross sections with the s, p polarizations, as well as their sums, are

```
In[63] := {Plot[{σ0s[ϑ], σ1s[ϑ], σ2s[ϑ], σ0s[ϑ] + σ1s[ϑ] + σ2s[ϑ]}, {ϑ, 0, π},
        PlotRange->{0, 1}, PlotStyle->col],
    Plot[{σ0p[ϑ], σ1p[ϑ], σ2p[ϑ], σ0p[ϑ] + σ1p[ϑ] + σ2p[ϑ]}, {ϑ, 0, π},
        PlotRange->{0, 1}, PlotStyle->col]}
```

Out[63] =



Of course, there are also higher contributions, not included here. The cross section is small for scattering angles below the rainbow peak; i.e., the sky just outside the rainbow is darker.

Near the rainbow peak, the last contribution (2 ray segments) is dominant; the ratio of the s and p polarizations is

```
In[64] := Rs2 = Simplify[TrigExpand[Rs/. Snell/. ρ->ρ2r], n > 1];
    Rp2 = Simplify[TrigExpand[Rp/. Snell/. ρ->ρ2r], n > 1];
    P2 = Simplify[(1 - Rs2)^2 * Rs2/((1 - Rp2)^2 * Rp2), n > 1]
```

$$\text{Out[64]} = \frac{\left(2+n^2\right)^6}{729 n^4 \left(-2+n^2\right)^2}$$

```
In[65] := P2/. Water
```

Out[65] = 25.3347

The rainbow light is highly linearly polarized, with the electric field orthogonal to the scattering plane, i.e., along the rainbow. The reason is that the incidence angle β of the ray which is about to reflect from the inner surface of the drop

```
In[66] := β/.Snell/.ρ−>ρ2r/.Water
Out[66] = 0.702055
```

is close to the Brewster angle

```
In[67] := βB/.Water
Out[67] = 0.643621
```

23.5 *L* Ray Segments Inside the Drop

Repeating the arguments from the previous sections, we obtain the direction of the outgoing ray $\vartheta L = 2(\beta - \alpha) - (L-1)(\pi - 2\beta)$:

```
In[68] := ϑL = 2∗L∗β − 2∗α − (L − 1)∗π;
In[69] := Ray[L_,y_,a_] := With[{α = α/.Snell/.ρ−>y,
      β = β/.Snell/.ρ−>y/.Water},
   Module[{R = {col[[1]],Line[{{−1 − a,y},{−Sqrt[1 − y^2],y}}]},
      φ = π − α,φ1,ϑ = 2∗L∗β − 2∗α − (L − 1)∗π},
    Do[φ1 = φ + 2∗β − π;
      R = Join[R,{col[[m + 2]],Line[{{Cos[φ],Sin[φ]},{Cos[φ1],Sin[φ1]}}]}];
      φ = φ1,{m,L}];
    Join[R,{col[[2]],
      Line[{{Cos[φ],Sin[φ]},{Cos[φ] + a∗Cos[ϑ],Sin[φ] + a∗Sin[ϑ]}}]}]]]
In[70] := ϑ[L_,ρ_] = ϑL/.Snell/.Water;
In[71] := Manipulate[{
   Manipulate[
    Graphics[Join[{Black,Circle[]},Ray[L,y,a]],
      PlotRange−>{{−1 − a,1 + a},{−1 − a,1 + a}}],
    {{y,0.6},0,1},{{a,2},1,10}],
   Manipulate[
    Graphics[Join[{Black,Circle[]},
      With[{δ = (ρmax − ρmin)/M},
        If[δ > 0,Apply[Join,Table[Ray[L,y,a],{y,ρmin + δ/2,ρmax,δ}]],{}]]],
      PlotRange−>{{−1 − a,1 + a},{−1 − a,1 + a}}],
    {{ρmin,0},0,1},{{ρmax,1},0,1},
    {{M,10},Table[i,{i,30}]},{{a,2},1,10}],
   Plot[Abs[Mod[ϑ[L,ρ],2∗π,−π]],{ρ,0,1}]},
   {{L,3},Table[i,{i,0,4}]}]
```

Out[71] =



Scattering angles for $L = 0, 1, 2, 3, 4$ as functions of ρ are

**In[72] := Plot[Evaluate[Table[Abs[Mod[ϑ[L,ρ],2 * π, −π]],{L,0,4}]],{ρ,0,1},
PlotStyle−>col]**

Out[72] =



Naturally, those with even L start from π at $\rho = 0$; for odd L they start from 0. For $L \geqslant 2$ each one has a single extremum:

In[73] := sol = Solve[D[ϑL/.Snell, ρ] == 0, ρ]

$$\text{Out}[73] = \left\{\left\{\rho \to -\frac{\sqrt{L^2-n^2}}{\sqrt{-1+L^2}}\right\}, \left\{\rho \to \frac{\sqrt{L^2-n^2}}{\sqrt{-1+L^2}}\right\}\right\}$$

In[74] := ρL = ρ/.sol[[2]]

$$\text{Out}[74] = \frac{\sqrt{L^2-n^2}}{\sqrt{-1+L^2}}$$

In[75] := ϑr = ϑL/.Snell/.ρ->ρL

$$\text{Out}[75] = -(-1+L)\pi - 2\,\text{ArcSin}\left[\frac{\sqrt{L^2-n^2}}{\sqrt{-1+L^2}}\right] + 2L\,\text{ArcSin}\left[\frac{\sqrt{L^2-n^2}}{\sqrt{-1+L^2}n}\right]$$

These extrema produce rainbows at the angles (in degrees)

In[76] := Table[(π − Abs[Mod[ϑr/.Water, 2 ∗ π, −π]])/Degree, {L, 2, 4}]

$\text{Out}[76] = \{42.0781, 50.8908, 138.263\}$

The angles at which an observer sees the first and the second rainbow are

In[77] := {ϑr2, ϑr3} = {π + (ϑr/.L−>2), −π − (ϑr/.L−>3)}

$$\text{Out}[77] = \left\{-2\,\text{ArcSin}\left[\frac{\sqrt{4-n^2}}{\sqrt{3}}\right] + 4\,\text{ArcSin}\left[\frac{\sqrt{4-n^2}}{\sqrt{3}n}\right],\right.$$
$$\left.\pi + 2\,\text{ArcSin}\left[\frac{\sqrt{9-n^2}}{2\sqrt{2}}\right] - 6\,\text{ArcSin}\left[\frac{\sqrt{9-n^2}}{2\sqrt{2}n}\right]\right\}$$

The sky is somewhat darker between the first rainbow and the second one, because neither rays with $L = 2$ nor those with $L = 3$ come from these directions.

Until now, we discussed monochromatic light. Then each rainbow is just a bright arc of the same color. In fact, the refraction index of water n depends on the color (wavelength) of light (dispersion). It is larger for violet light than for red one. Therefore the positions of the maxima of intensity of the scattered light also depend on the color.

In[78] := Plot[{ϑr2/Degree, ϑr3/Degree}, {n, 1.325, 1.335}, PlotStyle−>col]

Out[78] =



We see that the order of colors in the second rainbow is opposite to that in the first one; and the second rainbow is wider.

The ratios of s and p polarizations at the rainbow peaks are

In[79] := Rs0 = Simplify[TrigExpand[Rs/. Snell/. ρ−>ρL], {*n* > 1, *L* > 1}]

$$\text{Out[79]} = \frac{(-1+L)^2}{(1+L)^2}$$

In[80] := Rp0 = Simplify[TrigExpand[Rp/. Snell/. ρ−>ρL], {*n* > 1, *L* > 1}]

$$\text{Out[80]} = \frac{\left(L-n^2\right)^2}{\left(L+n^2\right)^2}$$

In[81] := *P* = Simplify[(1 − Rs0)^2 ∗ Rs0^(*L* − 1)/((1 − Rp0)^2 ∗ Rp0^(*L* − 1)), {*n* > 1, *L* > 1}]

$$\text{Out[81]} = \frac{\left(\frac{(-1+L)^2\left(L+n^2\right)^2}{(1+L)^2\left(L-n^2\right)^2}\right)^L \left(L^2-n^4\right)^2}{\left(-1+L^2\right)^2 n^4}$$

In[82] := Table[*P*/. Water, {*L*, 2, 4}]

Out[82] = {25.3347, 9.36736, 8.10737}

Higher rainbows are not so strongly polarized as the first one, because the incidence angle for the reflections inside the drop is not so close to the Brewster angle.